

Лабораторна робота №4

Використання сучасних інтегрованих АСУП на прикладі 1С:Підприємство

Даний процес автоматизує послідовність видачі доручень (на проведення будь-якої дії). Наприклад, хтось повинен замовити авіаквитки для директора (поїздки у відрядження), хтось цей процес повинен проконтролювати і т.п.

У бізнес-процесу визначені наступні реквізити (табл. 1).

Таблиця 1. Реквізити бізнес-процесу «Поручение»

Реквизит	Тип
Автор	СправочникСсылка.Сотрудники
Исполнитель	СправочникСсылка.Сотрудники
Исполнитель Роль	СправочникСсылка.РолиИсполнителей
ИсполнительОтдел	СправочникСсылка.Подразделения
Контролер	СправочникСсылка.Сотрудники
КонтролерРоль	СправочникСсылка.РолиИсполнителей
КонтролерОтдел	СправочникСсылка.Подразделения
Приоритет	Число
Тема	Строка
ТекстПоручения	Строка
ОтчетОВыполнении	Строка
ОтчетОКонтроле	Строка
Параллельный Контроль	Булево
ДатаВыполнения	Дата
ДатаКонтроля	Дата
КонтрольныйСрок	Дата
ОтметкаОВыполнении	Булево
ОтметкаОСнятииСКонтроля	Булево

Карта бізнес-процесу має такий вигляд (рис. 1).

У даного бізнес-процесу визначено дві точки старту. Спроба запуснути процес штатними засобами призведе до помилки (не визначено точку старту). Забігаючи наперед, скажемо, що даний бізнес-процес буде використовуватися як вкладений (в цьому випадку і буде використовуватися точка старту СтартБезАвтора). Самостійний старт буде відбуватися з точки Старт. Як іншу особливість даного бізнес-процесу можна відзначити той факт, що у всіх точках дії не визначено ні одна з властивостей, відповідальних за адресацію (ці властивості визначаються програмно).




Рис. 1. Карта маршруту бізнес-процесу

Старт бізнес-процесу можна реалізовувати різними способами. Розглянемо варіант, коли в модулі спільної команди визначається процедура (рис. 2)

```

□ Процедура СтартВП ()
    БизнесПроцесс = БизнесПроцессы.Поручение.СоздатьБизнесПроцесс ();
    БизнесПроцесс.Дата = ТекущаяДата ();
    БизнесПроцесс.Записать ();
    БизнесПроцесс.Старт (БизнесПроцессы.Поручение.ТочкиМаршрута.Старт );
    КонецПроцедуры
  
```

Рис. 2. Процедура, яка виконує старт бізнес-процесу

У самій процедурі уточнюється точка старту, і бізнес-процес стартує з цієї точки.

При старті виконується наступний обробник точки старту бізнес-процесу (рис. 3).

```

□ Процедура СтартПередСтартом(ТочкаМаршрутаБизнесПроцесса, Отказ)
    Автор = ПараметрыСеанса.ТекущийИсполнитель;
    КонтрольныйСрок = КонецНедели(ТекущаяДата());
    Записать();

    Сообщение = Новый СообщениеПользователю();
    Сообщение.Текст = "Старт бизнес-процесса: " + "" + ЭтотОбъект.Метаданные().Синоним + "";
    Сообщение.Сообщить();
    КонецПроцедуры
  
```

Рис. 3. Обработка події «Перед стартом»

У ньому до реквізиту бізнес-процесу Автор записується значення з параметра сеансу, що містить поточного користувача (елемент довідника). Встановлюється контрольний термін виконання.

Після старту бізнес-процес переходить до точки дії Видати доручення. Слід зазначити, що в даній точці відключений механізм створення завдань за замовчуванням. У ній визначено обробник події Перед созданием задач (рис. 4).

```

□ Процедура ВыдатьПоручениеПередСозданиемЗадач (ТочкаМаршрутаБизнесПроцесса, формируемьеЗадачи, СтандартнаяОбработка)

    Задача = Задачи.Задача.СоздатьЗадачу ();
    Задача.БизнесПроцесс = Ссылка;
    Задача.ТочкаМаршрута = ТочкаМаршрутаБизнесПроцесса;
    Задача.Дата = ТекущаяДата ();
    Задача.Сотрудник = Автор;
    Задача.Наименование = ТочкаМаршрутаБизнесПроцесса.НаименованиеЗадачи;
    Если Не ПустаяСтрока(Тема) Тогда
        Задача.Наименование = Задача.Наименование + " : " + Тема;
    КонецЕсли;
    формируемьеЗадачи.Добавить(Задача);
    СтандартнаяОбработка = Ложь;

КонецПроцедуры

```

Рис. 4. Обробник події «ПередСозданиемЗадач»

Як значення реквізиту адресації завдання встановлюється значення з реквізиту бізнес-процесу Автор. Стандартний механізм створення завдань відключається.

При активації даної задачі виконується наступний код обробника події ОбработкаПолученияФорми модуля менеджера завдання (рис. 5).

```

□ Процедура ОбработкаПолученияФормы(ВидФормы, Параметры, ВыбраннаяФорма, ДополнительнаяИнформация, СтандартнаяОбработка)

    Если ВидФормы="формаОбъекта" Тогда
        ЗадачаСсылка=Параметры.Ключ;
        ЗадачаОбъект=ЗадачаСсылка.ПолучитьОбъект ();

        Если ЗадачаОбъект.ТочкаМаршрута=БизнесПроцессы.Поручение.ТочкиМаршрута.ВыдатьПоручение Тогда
            ВыбраннаяФорма="БизнесПроцессы.Поручение.форма.формаПоручения";
            Параметры.Ключ=ЗадачаОбъект.БизнесПроцесс;
            Параметры.Вставить("Задача", ЗадачаСсылка);
            СтандартнаяОбработка=Ложь;
        КонецЕсли;
    КонецЕсли;

КонецПроцедуры

```

Рис. 5. Обробник події «ОбработкаПолученияФормы»

При активації завдання відкривається форма бізнес-процесу ФормаПоручения - специфічна форма бізнес-процесу, в якій кожен з учасників процесу може залишати будь-які позначки, зауваження. У ній реалізовано розмежування прав доступу до різних полях (реквізитами бізнес-процесу) залежно від ролі співробітника. Наприклад, виконавець не може виправити текст доручення тощо

При створенні форми на сервері в реквізит форми ТекущаяЗадача в параметрі форми передається поточна (активується) завдання (рис. 6).

```

□ Процедура ПриСозданииНаСервере (Отказ, СтандартнаяОбработка)

    ТекущаяЗадача = Параметры.Задача;
    Если ТекущаяЗадача.Выполнена Тогда
        Элементы.ВыполнитьЗадачу.Доступность = Ложь;
    КонецЕсли;

КонецПроцедуры

```

Рис. 6. Обробник події форми «ПриСозданииНаСервере»

Оскільки в бізнес-процесах Поручение і Согласование при активації завдання замість форми завдання відкриваються спеціалізовані форми бізнес-процесів, для виконання цього завдання у формах існує команда

ВыполнитьЗадачу і відповідна їй кнопка. Ця кнопка доступна в разі, якщо завдання не виконано. При її натисканні завдання виконується (рис. 7).

```

Процедура ВыполнитьТекущуюЗадачу ()
    //Если ТекущаяЗадача.ПроверитьВыполнение () Тогда
        ТекущаяЗадача.ПолучитьОбъект () .ВыполнитьЗадачу ();
    //КонецЕсли;
КонецПроцедуры

```

Рис. 7. Виконання поточної задачі

При необхідності перед виконанням завдання можна перевірити умови її виконання методом ПроверитьВыполнение ().

Надалі цей механізм виконання завдань буде використовуватися в інших завданнях бізнес-процесів Поручение і Согласование.

Перед виконанням завдання Выдача поручений, створеної в точці Видати доручення, необхідно перевірити: чи не порожній чи текст доручення, чи визначено його виконавець. Дані перевірки виконуються в обробнику події Перед выполнением. Текст обробника наведено нижче (рис. 8).

```

Процедура ВыдатьПоручениеПередВыполнением(ТочкаМаршрутаБизнесПроцесса, Задача, Отказ)
    Отказ = Ложь;
    // Проверяется заполнение хотя бы одного реквизита, который будет использоваться в качестве значений
    // реквизита адресации задачи
    Если Исполнитель.Пустая() И ИсполнительРоль.Пустая() И ИсполнительОтдел.Пустая() Тогда
        Отказ = Истина;
    КонецЕсли;
    Если ПустаяСтрока(ТекстПоручения) Тогда
        Отказ = Истина;
    КонецЕсли;
КонецПроцедуры

```

Рис. 8. Обробник події «ПередВыполнением»

В ідеальному випадку таку ж перевірку необхідно було б реалізувати і для співробітника, що здійснює контроль виконання доручення.

Після виконання завдання бізнес-процес переходить до точки перевірки умови. У ній проводиться вибір варіанту виконання бізнес-процесу. Вибір здійснюється у відповідності зі значенням реквізиту бізнес-процесу ПараллельныйКонтроль (встановленим інтерактивно). Оброблювач перевірки умови містить фактично один рядок (рис. 9)

```

Процедура ПараллельныйКонтрольПроверкаУсловия (ТочкаМаршрутаБизнесПроцесса, Результат)
    Результат = ПараллельныйКонтроль;
КонецПроцедуры

```

Рис. 9. Обробник події «ПроверкаУсловия»

Припустимо (для початку), що був обраний варіант з паралельним контролем (тобто контролер може встановити позначку про успішність/неуспішність результатів виконання раніше, ніж сам виконавець зареєструє факт виконання). При такому варіанті бізнес-процес розпаралелюється і одночасно переходить до точок дії Виконати доручення і Контроль виконання. У кожній з них формування завдань проводиться програмно в обробнику події Перед созданием задач. Текст обробника для точки дії Выполнить поручение має наступний (рис. 10).

```

Процедура ВыполнитьПоручениеПередСозданиемЗадач (ТочкаМаршрутаБизнесПроцесса, формируемыеЗадачи, СтандартнаяОбработка)

    Задача = Задачи.Задача.СоздатьЗадачу();
    Задача.БизнесПроцесс = Ссылка;
    Задача.ТочкаМаршрута = ТочкаМаршрутаБизнесПроцесса;
    Задача.Дата = ТекушаяДата();
    Задача.Приоритет = Приоритет;
    Задача.КонтрольныйСрок = КонтрольныйСрок;
    Если Не Исполнитель.Пустая() Тогда
        Задача.Сотрудник = Исполнитель;
    КонецЕсли;

    Если Не ИсполнительРоль.Пустая() Тогда
        Задача.РольИсполнителя = ИсполнительРоль;
    КонецЕсли;

    Если Не ИсполнительОтдел.Пустая() Тогда
        Задача.Подразделение = ИсполнительОтдел;
    КонецЕсли;
    Задача.Наименование = ТочкаМаршрутаБизнесПроцесса.НаименованиеЗадачи;
    Если Не ПустаяСтрока(Тема) Тогда
        Задача.Наименование = Задача.Наименование + ": " + Тема;
    КонецЕсли;
    формируемыеЗадачи.Добавить(Задача);
    СтандартнаяОбработка = Ложь;
КонецПроцедуры

```

Рис. 10. Обробник події «ПередСозданиемЗадач»

В якості значень реквізитів адресації завдання встановлюються значення з відповідних реквізитів бізнес-процесів. Обробник події Перед созданием задач для точки дії Контроль виконання має такий же код, за винятком того, що реквізити адресації завдання заповнюються значеннями реквізитів бізнес-процесу, пов'язаних із співробітником, який виконує контроль.

При активації даної задачі виконується наступний код обробника події ОбработкаПолученияФормы модуля менеджера завдання (рис. 11).

```

□ Процедура ОбработкаПолученияФормы(ВидФормы, Параметры, ВыбраннаяФорма, ДополнительнаяИнформация, СтандартнаяОбработка)
    Если ВидФормы="ФормаОбъекта" Тогда
        ЗадачаСсылка=Параметры.Ключ;
        ЗадачаОбъект=ЗадачаСсылка.ПолучитьОбъект();

        Если ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.ВыполнитьПоручение Или
           ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.ВыполнитьПоручение2 Или
           ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.КонтрольИсполнения Или
           ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.КонтрольИсполнения2 Тогда
            ВыбраннаяФорма="БизнесПроцессы.Поручение.Форма.ФормаПоручения";
            Параметры.Ключ=ЗадачаОбъект.БизнесПроцесс;
            Параметры.Вставить("Задача", ЗадачаСсылка);
            СтандартнаяОбработка=Ложь;

        КонецЕсли;
    КонецЕсли;
КонецПроцедуры

```

Рис. 11. Обробник події «ОбработкаПолученияФормы»

Її виконання призводить до відкриття форми ФормаПоручения, форма завдання при цьому не відкривається.

Виконання завдання здійснюється за допомогою посилання на задачу в реквізиті форми ТекущаяЗадача і команди ВыполнитьЗадачу.

Після того як обидва завдання будуть виконані, бізнес-процес пройде точку злиття і перейде до точки перевірки умови Знято з контролю? У обробнику перевірки умови перевіряється значення реквізиту ОтметкаОснятииСКонтроля (встановлюється інтерактивно у формі ФормаПоручения співробітником, що здійснює контроль, і означає, що доручення виконано). Якщо доручення не виконане, проводиться повернення до точки перевірки умови Паралельний контроль?, в іншому випадку бізнес-процес завершується (здійснюється перехід до точки завершення).

Розглянемо поведінку бізнес-процесу в разі, коли в точці перевірки умови Паралельний контроль? виконання бізнес-процесу піде по гілці Ні.

Відповідно з картою маршруту бізнес-процес переходить до точки дії Виконати доручення (2).

Обробник події Перед созданием задач для точки дії Виконати доручення використовується такий же, як і у точки Виконати доручення при виконанні паралельного контролю (рис. 12).

```

Процедура ВыполнитьПоручениеПередСозданиемЗадач(ТочкаМаршрутаБизнесПроцесса, формируемыеЗадачи, СтандартнаяОбработка)
    Задача = Задачи.Задача.СоздатьЗадачу();
    Задача.БизнесПроцесс = Ссылка;
    Задача.ТочкаМаршрута = ТочкаМаршрутаБизнесПроцесса;
    Задача.Дата = ТекущаяДата();
    Задача.Приоритет = Приоритет;
    Задача.КонтрольныйСрок = КонтрольныйСрок;
    Если Не Исполнитель.Пустая() Тогда
        Задача.Сотрудник = Исполнитель;
    КонецЕсли;

    Если Не ИсполнительРоль.Пустая() Тогда
        Задача.РольИсполнителя = ИсполнительРоль;
    КонецЕсли;

    Если Не ИсполнительОтдел.Пустая() Тогда
        Задача.Подразделение = ИсполнительОтдел;
    КонецЕсли;
    Задача.Наименование = ТочкаМаршрутаБизнесПроцесса.НаименованиеЗадачи;
    Если Не ПустаяСтрока(Тема) Тогда
        Задача.Наименование = Задача.Наименование + ": " + Тема;
    КонецЕсли;
    формируемыеЗадачи.Добавить(Задача);
    СтандартнаяОбработка = Ложь;
КонецПроцедуры

```

Рис. 12. Обробник події «ПередСозданиемЗадач»

В якості значень реквізитів адресації завдання встановлюються значення з відповідних реквізитів бізнес-процесів.

При активації викликається та ж процедура, що визначена для точки Виконати доручення при виконанні паралельного контролю (рис. 13).

```

Процедура ОбработкаПолученияФормы(ВидФормы, Параметры, ВыбраннаяФорма, ДополнительнаяИнформация, СтандартнаяОбработка)
    Если ВидФормы="формаОбъекта" Тогда
        ЗадачаСсылка=Параметры.Ключ;
        ЗадачаОбъект=ЗадачаСсылка.ПолучитьОбъект();

        Если ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.ВыполнитьПоручение Или
            ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.ВыполнитьПоручение2 Или
            ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.КонтрольИсполнения Или
            ЗадачаОбъект.ТочкаМаршрута = БизнесПроцессы.Поручение.ТочкиМаршрута.КонтрольИсполнения2 Тогда
            ВыбраннаяФорма="БизнесПроцессы.Поручение.форма.формаПоручения";
            Параметры.Ключ=ЗадачаОбъект.БизнесПроцесс;
            Параметры.Вставить("Задача", ЗадачаСсылка);
            СтандартнаяОбработка=Ложь;
        КонецЕсли;
    КонецЕсли;
КонецПроцедуры

```

Рис. 13. Обробник події «ОбработкаПолученияФормы»

Її виконання призводить до відкриття форми ФормаПоручения, форма завдання при цьому не відкривається.

Виконання завдання здійснюється за допомогою посилання на задачу в реквізиті форми ТекущаяЗадача і команди ВыполнитьЗадачу.

Після виконання завдання бізнес-процес переходить до точки перевірки умови Потрібен контроль? Оброблювач перевірки умови містить наступний текст (рис. 14).

```

□ Процедура НуженКонтрольПроверкаУсловия(ТочкаМаршрутаБизнесПроцесса, Результат)
    Результат = Истина;
    Если Контролер.Пустая() И КонтролерРоль.Пустая() И КонтролерОтдел.Пустая() Тогда
        Результат = Ложь;
    КонечЕсли;
КонецПроцедуры

```

Рис. 14. Обробник події «ПроверкаУсловия»

У процедурі проводиться перевірка на наповненість будь-якого реквізиту, значення якого може використовуватися в якості значення реквізитів адресації завдання. Якщо всі відповідні реквізити містять порожні посилання, бізнес-процес переходить по гілки Ні до свого завершення. Якщо виконання пішло по гілки Так, відповідно до карти бізнес-процесу проводиться перехід до точки дії Контроль виконання. Обробники подій у даної точки описані таким же чином, як і в точці Контроль виконання при паралельному варіанті проведення контролю.

Після виконання завдання, створеної в точці маршруту Контроль виконання, бізнес-процес переходить до точки перевірки умови Знято з контролю? У разі успішного виконання доручення (критерії були описані раніше) бізнес-процес завершується.